構造解析学の学習に数式処理ソフトを使う

斉木　功

1　数式処理ソフト Maxima を使う

1.1　Maxima の導入

Maxima とは，微分や積分，線形連立方程式，多項式，ベクトルや行列などを含む記号や数値による表現を処理するフリーソフトウェアである．このようなソフトウェアとしては，Mathematica (Wolfram 社) が有名であり，著者としても使い慣れているが，少々高価なので，気軽に利用することのできる Maxima を構造解析の学習に利用する方法をここで述べることとする．

Maxima の生い立ちなどについては，オフィシャルなウェブサイトを見てもらうこととして，以下ではこれを利用する最低限の手続きについて述べる．まず，使用するコンピュータの OS は Windows であれば，Windows 用の Maxima の名前である wxMaxima で検索をすると，たとえば

`http://sourceforge.net/projects/maxima/files/`

あたりがヒットするだろう．Maxima-Windows なるキーワードから，インストール用の実行ファイルをダウンロードし，実行すればインストールは難しくないだろう．

インストールができたら，さっそく起動してみよう．起動すると地味な白地の窓があらわれる．あまり意味はないが，手始めにこのウィンドウに `1+2;` と入力し，最後に Shift + Enter を押してみよう．Shift + Enter は Shift キーを押しながら Enter キーを押すという意味である．すると `3` という当然の答えが返ってくる．このように，Maxima では単なる Enter ではなく，Shift + Enter が入力した情報を処理せよ，という意味になっている．これ以降は，Shift + Enter の表記は省略する．なお，単なる Enter は入力を見やすくするために改行するのに使える．例えば，決して見やすくはないが，`1` Enter `+2;` と入力しても，結果は同じ `3` となる．また，`;`（セミコロン）は，入力の意味の区切りを意味する．例えば，`1+2; 3+4;` とすれば，結果は `3` と `7` が返ってくる．

上の例はあまり感動はしないだろうが，次は `1/2+1/3;` と入力してみよう．すると，$\frac{5}{6}$ という答えが返ってくる．これは少し感動してもらえただろうか．著者も学生のときに，関数電卓で分数計算ができることに感動し，構造解析の勉強に大いに活用した記憶がある．

Maxima の基本的な使い方については，多くのウェブサイトがあるので，あまり深入りするつもりはない．最後に `(x+3*y)+(2*x-y);` と入力してみよう．すると，`2y+3x` と出力されるだろう．このように，数式を記号のまま処理できるのが数式処理ソフトというものである．なお，出力では乗算の記号は省かれているが，入力では `*` を省くことはできない．

1.2　Maxima による作図と関数の定義

梁の境界値問題をいきなり解く（解かせる）ことができればよいが，少し説明が長くなるので，まずは構造解析で求めた曲げモーメント図やせん断力図，あるいはたわみ曲線といった図を Maxima で確認することからはじめよう．例えば，図 1 に示す長さ ℓ の単純支持梁に等分布荷重 q_0 が作用しているおなじみの問題では，曲げモーメント M は $M(x)=\frac{q_0\ell^2}{2}\left\{\left(\frac{x}{\ell}\right)-\left(\frac{x}{\ell}\right)^2\right\}$ となる．数式処理ソフトでも，P や ℓ といった記号のままグラフを描くことはできないので，無次元化した曲げモーメント $\overline{M}(x)=\frac{M}{q_0\ell^2}=\frac{1}{2}\left\{\left(\frac{x}{\ell}\right)-\left(\frac{x}{\ell}\right)^2\right\}$ を図化してみよう．座標系も，無次元化した座標 $\overline{x}=\frac{x}{\ell}$ を使う．では

```
plot2d(1/2*(x-x^2),[x,0,1]);
```

と入力してみよう．これは，関数 $\frac{1}{2}\left(x-x^2\right)$ を $0\leq x\leq 1$ の範囲で描けという意味である．すると，別なウィンドウが開いて図 2 に示すようなグラフがあらわれる．なお，このグラフのウィンドウを閉じないと，Maxima での他の作業ができないようである．このグラフは，通常の数学のグラフと同様，上向きが正になっている．構造解析において，曲げモーメントやたわみは下向きに正にすることが多いので，グラフを描く時だけは関数の符号を逆にするとよいかもしれない．

次に，結果は同じであるが，曲げモーメントを `M` という関数に定義してからグラフ化してみよう．関数を定義するには `:=` を使う．上と同じ単純支持梁の曲げモーメントは `M(x):=1/2*(x-x^2);` と定義する．この，定義された関数を用いて，符号を逆にしたグラフを描くには `plot2d(-M(x),[x,0,1]);` と入力する．大文字と小文字は区別するので注意しよ



図 1: 等分布荷重を受ける単純支持梁



図 2: 等分布荷重を受ける単純支持梁の曲げモーメント分布（上向き正）

う．一度関数として定義してしまえば，例えば $x = 0$ や $x = \frac{\ell}{2}$ における曲げモーメントを確認するためには `M(0);` や `M(1/2);` と入力すればよい．今のケースでは，それぞれ 0 と $\frac{1}{8}$ と出力されるだろう．これらは無次元化されているので，$x = \frac{\ell}{2}$ における実際の曲げモーメントは $\frac{1}{8} \times q_0 \ell^2$ である．

先ほどの例と違い，集中荷重がある場合は曲げモーメントやたわみは場合分けが必要になる．そのようなときも，`if` を用いると一つの関数として定義できる．例えば，先ほどと同じ長さ ℓ の単純支持梁の中央に集中荷重 P が作用している場合，曲げモーメントは

$$M(x) = \begin{cases} P\ell \frac{x}{2\ell} & (x \leq \frac{\ell}{2}) \\ P\ell \left(\frac{1}{2} - \frac{x}{2\ell}\right) & (\frac{\ell}{2} \leq x) \end{cases} \tag{1}$$

となる．無次元化を考慮して `M(x):=if x<1/2 then x/2 else 1/2 - x/2;` としてから `plot2d(-M(x), [x,0,1]);` としてみよう．おなじみの V 字形の曲げモーメント分布が確認できるだろう．関数 `M` は `if` の後の条件 `x<1/2` が成立する範囲では `then` 以下の `x/2` により定義され，条件が満たされないときは `else` 以下の `1/2 - x/2` により定義される．

1.3　Maxima による梁の境界値問題の解法

Maxima には非常にたくさんの機能があるが，ここではとにかく梁の境界値問題を解くことを考える．もちろん，以下に示す方法以外にもやり方はあると思うので，興味のある人はいろいろと試してみよう．

1.3.1　等分布荷重が作用する場合

まず，微分方程式の一般解（たわみ）を定義する．長さ ℓ の梁に等分布荷重 q_0 が作用しているとし，梁の曲げ剛性も EI で一定とすれば，たわみの微分方程式

$$\frac{\mathrm{d}^4 v}{\mathrm{d}x^4} = \frac{q_0}{EI} \tag{2}$$

より，一般解は

$$v(x) = \frac{q_0 \ell^4}{24EI} \left\{ \overline{x}^4 + c_1 \overline{x}^3 + c_2 \overline{x}^2 + c_3 \overline{x} + c_4 \right\} \tag{3}$$

と表せる．ここに，$\overline{x} = \frac{x}{\ell}$ は無次元化した座標である．無次元化したたわみ $\overline{v} = \frac{24vEI}{q_0 \ell^4}$ を `v1(x)` とすると，Maxima では

```
v1(x):=x^4+c1*x^3+c2*x^2+c3*x+c4;
```

と定義すればよいだろう．

つぎに，境界条件の設定で必要になる可能性のある，たわみ角，曲げモーメント，せん断力を定義する．ただし，たわみを無次元化したことを考慮し，たわみ角 `th1(x)`，曲げモーメント `M1(x)`，せん断力 `Q1(x)` はそれぞれ無次元化たわみの 1 階，2 階，3 階微分とする．これらを Maxima で定義するには



図 3: 等分布荷重を受ける片持梁



図 4: 等分布荷重を受ける片持梁のたわみと曲げモーメント分布（下向き正）

```
define(th1(x),-diff(v1(x),x,1));
define(M1(x),-diff(v1(x),x,2));
define(Q1(x),-diff(v1(x),x,3));
```

とする．ここで，`define(f(x), expr)` は関数 `f(x)` を `expr` で定義をするための命令である．最初のたわみの定義で使った`:=`を使っていないのは理由があるが，詳しくは Maxima 関連の文献やウェブサイトを参照してほしい．`diff(f(x),x,n)` は，関数 `f(x)` の変数 `x` に関する `n` 階微分を意味する．

ようやく境界条件を用いて未定定数を求める準備が整った．具体的な問題として，図 3 に示す左端が埋め込み支持，右端が自由の片持梁を考えよう．このとき，境界条件は

$$v(0) = 0, \quad \theta(0) = 0, \quad M(\ell) = 0, \quad Q(\ell) = 0 \tag{4}$$

である．座標が無次元化されていることを考慮し，Maxima でこの連立方程式を解くために

```
sol:solve([v1(0)=0,th1(0)=0,M1(1)=0,Q1(1)=0],[c1,c2,c3,c4]);
```

としてみよう．すると，各未定定数の解が出力されるだろう．解を求めるだけであれば，上記の `sol:`は必要ない．これは，`sol` という要素（変数といってもいいだろう）に:以下，すなわち `solve` の結果を割り当てる（代入する）という意味である．これは，求まった未定定数を代入して，たわみを図化したい場合に役に立つ．例えば

```
plot2d(-ev(v1(x),sol),[x,0,1]);
```

と入力してみよう．すると，境界条件を考慮した上で求まったたわみの形がグラフとして描かれるだろう．つまり，`ev(v1(x),sol)` は，`sol` に割り当てられた未定定数の解を一般解 `v1(x)` に代入した関数を意味する．曲げモーメント図を確認したければ，同様に

```
plot2d(-ev(M1(x),sol),[x,0,1]);
```

などとすればよいだろう．たわみ・曲げモーメントの `plot2d` の結果を図 3 に示す．具体的なある点でのたわみや曲げモーメントを確認したい場合は，`v1` や `M1` が無次元化されていることに注意して，例えば右端でのたわみは

```
ev(v1(1)*q0*L^4/(24*E*I),sol);
```



図 5: 集中荷重を受ける片持梁



図 6: 集中荷重を受ける片持梁のたわみと曲げモーメント分布（下向き正）

とすれば，$\frac{q_0\ell^4}{8EI}$ であることが確認できる．また，左端での曲げモーメントは

```
ev(M1(0)*q0*L^2/24,sol);
```

とすれば $\frac{-q_0\ell^2}{2}$ であることがわかる．

1.3.2 集中荷重が作用する場合

等分布荷重のときと同様に，まず，微分方程式の一般解（たわみ）を定義する．長さ ℓ の梁に等分布荷重が作用しておらず，集中荷重 P が作用しているとし，梁の曲げ剛性は EI で一定とすれば，たわみの微分方程式

$$\frac{\mathrm{d}^4 v}{\mathrm{d}x^4} = \frac{0}{EI} \tag{5}$$

より，一般解は

$$v(x) = \frac{P\ell^3}{EI}\left\{c_1\overline{x}^3 + c_2\overline{x}^2 + c_3\overline{x} + c_4\right\} \tag{6}$$

と表せる．ここに，$\overline{x} = \frac{x}{\ell}$ は無次元化した座標である．無次元化したたわみ $\overline{v} = \frac{vEI}{P\ell^3}$ を `v1(x)` とすると，Maxima では

```
v1(x):=c1*x^3+c2*x^2+c3*x+c4;
```

と定義すればよいだろう．つぎに，たわみ角，曲げモーメント，せん断力を定義する．分布荷重のときと同様に，無次元化を考慮し，たわみ角 `th1(x)`，曲げモーメント `M1(x)`，せん断力 `Q1(x)` をそれぞれ無次元化たわみの 1 階，2 階，3 階微分とする．これらを Maxima で

```
define(th1(x),-diff(v1(x),x,1));
define(M1(x),-diff(v1(x),x,2));
define(Q1(x),-diff(v1(x),x,3));
```

と定義する．

分布荷重のときと同様に，具体的な問題として，左端が埋め込み支持，右端が自由の片持梁を考えよう．図 5 に示すように集中荷重 P が右の自由端に作用しているとする．このとき，境界条件は

$$v(0) = 0, \quad \theta(0) = 0, \quad M(\ell) = 0, \quad Q(\ell) = P \tag{7}$$

である．座標が無次元化されていることを考慮し，Maxima でこの連立方程式を解くために

```
sol:solve([v1(0)=0,th1(0)=0,M1(1)=0,Q1(1)=1],[c1,c2,c3,c4]);
```

としてみよう．たわみの形を確認するには

```
plot2d(-ev(v1(x),sol),[x,0,1]);
```

曲げモーメント図の形を見たければ

```
plot2d(-ev(M1(x),sol),[x,0,1]);
```

などとすればよいだろう．たわみと曲げモーメントの `plot2d` の結果を図 6 に示した．無次元化に注意して，例えば右端でのたわみは

```
ev(v1(1)*P*L^3/(E*I),sol);
```

とすれば，$\frac{P\ell^3}{3EI}$ であることが確認できる．一方，左端での曲げモーメントは

```
ev(M1(0)*P*L/24,sol);
```

とすれば $-P\ell$ であることがわかる．

1.3.3 端部以外に集中荷重が作用する場合

ここでは，境界条件に加えて連続条件の考慮も必要になる場合の例として，梁の端部以外に集中荷重が作用する場合の解法をみてみよう．

これまでと同様に，まず，微分方程式の一般解（たわみ）を定義する．長さ ℓ の梁に等分布荷重が作用しておらず，集中荷重 P が中央に作用しているとし，梁の曲げ剛性は EI で一定とすれば，たわみの微分方程式の一般解は，$x \le \frac{\ell}{2}$ および $x \ge \frac{\ell}{2}$ においてそれぞれ

$$v_1(x) = \frac{P\ell^3}{EI}\left\{c_1\overline{x}^3 + c_2\overline{x}^2 + c_3\overline{x} + c_4\right\}, \quad v_2(x) = \frac{P\ell^3}{EI}\left\{c_5\overline{x}^3 + c_6\overline{x}^2 + c_7\overline{x} + c_8\right\} \tag{8}$$

と表せる．ここに，$\overline{x} = \frac{x}{\ell}$ は無次元化した座標である．無次元化したたわみ $\overline{v}_i = \frac{v_i EI}{P\ell^3}$ $(i = 1, 2)$ を `v1(x)`，`v2(x)` とすると，Maxima では

```
v1(x):=c1*x^3+c2*x^2+c3*x+c4;
v2(x):=c5*x^3+c6*x^2+c7*x+c8;
```

と定義すればよいだろう．つぎに，たわみ角，曲げモーメント，せん断力を

```
define(th1(x),-diff(v1(x),x,1));
define(M1(x),-diff(v1(x),x,2));
define(Q1(x),-diff(v1(x),x,3));
define(th2(x),-diff(v2(x),x,1));
define(M2(x),-diff(v2(x),x,2));
define(Q2(x),-diff(v2(x),x,3));
```

と定義する．

具体的な問題として，図 7 に示す単純支持両端単純支持の梁を考えよう．このとき，境界条件は

$$v_!(0) = 0, \quad M_1(0) = 0, \quad v_2(\ell) = 0, \quad M_2(\ell) = 0 \tag{9}$$

であり，連続条件は

$$v_1(\ell/2) = v_2(\ell/2), \quad \theta_1(\ell/2) = \theta_2(\ell/2), \quad M_1(\ell/2) = M_2(\ell/2), \quad Q_1(\ell/2) = Q_2(\ell/2) + P \tag{10}$$

である．座標が無次元化されていることを考慮し，Maxima でこの連立方程式を解くために



図 7: 集中荷重を受ける単純支持梁



図 8: 集中荷重を受ける単純支持梁のたわみと曲げモーメント分布（下向き正）

```
sol:solve([v1(0)=0,M1(0)=0,v2(1)=0,M2(1)=0,
           v1(1/2)=v2(1/2),th1(1/2)=th2(1/2),
           M1(1/2)=M2(1/2),Q1(1/2)=Q2(1/2)+1],
           [c1,c2,c3,c4,c5,c6,c7,c8]);
```

としてみよう．たわみの形を確認するには，まずはり全体を通してのたわみ `v(x)` を `if` を用いて

```
define(v(x),if x<1/2 then ev(v1(x),sol) else ev(v2(x),sol));
```

と定義した上で

```
plot2d(-v(x),[x,0,1]);
```

としてみよう．曲げモーメント図の形を見たければ

```
define(M(x),if x<1/2 then ev(M1(x),sol) else ev(M2(x),sol));
```

としてから

```
plot2d(-M(x),[x,0,1]);
```

などとすればよいだろう．たわみと曲げモーメントの `plot2d` の結果を図 8 に示す．無次元化に注意して，例えば集中荷重の載荷点（中央）でのたわみと曲げモーメントは

```
ev(v(1/2)*P*L^3/(E*I),sol);
ev(M(1/2)*P*L,sol);
```

とすれば，それぞれ $\frac{P\ell^3}{48EI}$ と $\frac{PL}{4}$ であることが確認できる．